Beate Stenstrøm skal købe 15 julegaver og grundet corona-epidemien, vil hun ordne det hele over nettet. Beate er et relativ nærigt menneske, og det er derfor vigtigt, at hun får de 15 julegaver, som hun har udset sig, så billigt som overhovedet muligt. Men hun må sande, at det ikke er helt trivielt at finde den billigste måde at købe gaverne på, da der er forsendelsesgebyrer ved de fleste forhandlere og priserne varierer også en del på tværs af online-butikkerne.

Priser på de 15 julegaver samt forsendelsesgebyrerne hos de seks forhandlerne hun har udset sig er givet i det vedlagte Excel-ark.

Beates plan er at bestille alle de gaver, som hun har planlagt at købe ved en given forhandler på én gang for på den måde at få så meget ud af forsendelsesgebyret som muligt. Det vil sige, at du kan antage, at hvis Beate vælger at købe mere end en gave ved en forhandler, skal forsendelsesgebyret kun betales én gang.

Der er desværre nogle forhandlere som ikke forhandler alle de gaver hun skal bruge. I Excel-arket er prisen for at købe en gave ved en forhandler, som ikke har denne til salg, sat til 100.000kroner for at indikere, at dette ikke er en mulighed.

Opgave 1:

Antag, at Beate bruger følgende grådige heuristik til at vælge hvor hun skal købe julegaverne: For hver gave vælger hun at købe den hos den forhandler, som har den laveste pris uden at skele til forsendelsesgebyret. Hvis flere forhandlere har den samme laveste pris, da vælger hun forhandleren med det lavest indeks. Hvad bliver omkostningerne for Beate ved at benytte denne strategi?

Svar: Beate køber sine 15 julegaver ved følgende forhandlere:

Gave	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15
Forretning	2	2	1	6	3	1	1	2	2	3	3	1	5	1	1
Pris	400	500	550	550	250	400	400	350	600	200	350	500	300	350	400

Dermed bruger Beate leverandørerne 1, 2, 3, 5 og 6. Udgifterne til gaverne beløber sig til 6.100kroner og forsendelsesgebyrerne løber op i 420 kroner. Samlet bliver Beates udgifter 6.520 kroner.

Givet din baggrund inden for prescriptive analytics ved du, at der ikke er nogen garanti for, at Beate har fundet en optimal løsning, da hun blot har brugt en heuristik til at finde en løsning. Derfor skal du i det følgende formulere en lineær heltalsmodel, som minimerer Beates samlede omkostninger ved at købe de 15 julegaver.

Du kan i det følgende vælge at bruge variablerne w_{fg}, som er defineret som følger

$$w_{fg} = \begin{cases} 1, hvis\ gave\ g = 1, \dots, 15\ købes\ ved\ forhandler\ f = 1, \dots, 6 \\ 0, ellers \end{cases}$$

Desuden kan du anvende variablerne z_f, som antager værdien 1 hvis en gave købes hos forhandler f og ellers er $z_f = 0$.

Opgave 2:

Formuler en objektfunktion, som minimerer de samlede omkostninger ved at købe gaver og betale for forsendelsesgebyrer.

Svar: Ved at lade c_{fg} være prisen ved at købe gave g ved forhandler f og ved at lade F_f være forsendelsesgebyret ved forhandler f kan objektfunktionen opskrives som

$$\min \sum_{f=1}^{6} \sum_{g=1}^{15} c_{fg} w_{fg} + \sum_{g=1}^{6} F_f z_f$$

Den første del af objektfunktionen dækker over de samlede udgifter til at købe gaverne mens den anden del summer forsendelsesgebyrerne som Beate bliver nødt til at betale.

Opgave 3:

Formuler begrænsninger som sørger for, at alle Beates 15 gaver bliver indkøbt.

Svar: For at sørge for at alle gaver bliver indkøbt, skal der vælges mindst en forhandler hvorfra gaven skal købes. Da hver gave har en strengt positiv pris, vil en gave aldrig købes mere end én gang, hvorfor vi kan formulere det som at der skal findes *præcis* en forhandler til hver gave. Dette kan gøres med følgende mængde af begrænsninger

$$\sum_{f=1}^{6} w_{fg} = 1, \quad \forall g = 1, \dots, 15$$

Opgave 4:

Formuler en mængde af *lineære* begrænsninger, som beskriver den logiske implikation

$$w_{fg} = 1 \Rightarrow z_f = 1, \quad \forall f = 1, \dots, 6 \;\; og \;\; g = 1, \dots, 15$$

Formuler et lineært heltalsprogram ved at kombinere objektfunktionen fra opgave 2, begrænsningerne fra opgave 3, samt begrænsningerne fra denne opgave.

Svar: Den logiske implikation kan, jf. dokumentet omkring modelleringstricks formuleres som følgende mængde af lineære uligheder

$$w_{fg} \leq z_f, \quad \forall f = 1, \dots, 6 \; og \; g = 1, \dots, 15$$

Her søger uligheden for, at hvis w_{fg} antager værdien 1 så skal $z_f = 1$ for at ulighederne er opfyldt, hvilket var målet.

Sammenfattes svarene fra Opgave 2 og 3 samt disse nye begrænsninger i en lineær heltalsmodel fås

$$\min \sum_{f=1}^{6} \sum_{g=1}^{15} c_{fg} w_{fg} + \sum_{g=1}^{6} F_f z_f$$

$$\text{s.t.:} \sum_{f=1}^{6} w_{fg} = 1, \quad \forall g = 1, \dots, 15$$

$$w_{fg} \leq z_f, \quad \forall f = 1, \dots, 6 \; og \; g = 1, \dots, 15$$

$$w_{fg} \in \{0,1\}, \quad \forall f = 1, \dots, 6 \; og \; g = 1, \dots 15$$

$$z_f \in \{0,1\}, \quad \forall f = 1, \dots, 6$$

Opgave 5:

Sammenlign modellen fra Opgave 4 med modeller gennemgået i løbet af kurset. Har Beates ”indkøbsmodel” nogen ligheder med modeller behandlet i kurset.

Svar: Modellen fra Opgave 4 er identisk med modellen for ”the uncapacitated facility location problem” UFLP. Variablerne w_{fg} svarer til allokeringsvariablerne x_{ij} i UFLP. Ligeledes svarer z_f variablerne til y_i variablerne i UFLP. Det vil sige, at gaverne som skal indkøbes for Beate tager rollen som kunder/efterspørgselspunkter og forhandlerne som skal udvælges tager rollen som ”facilities”.

Opgave 6:

Implementer modellen fra Opgave 4 i OPL og løs den ved hjælp af CPLEX. Dokumentér din implementering for eksempel ved hjælp af screenshots, kommentér på løsningen og sammenlign den med løsningen fundet i Opgave 1.

Svar: En implementering af modellen fra Opgave 4 kunne være

```
int n = ...;
int m = ...;
range gaver = 1..m;
range forretninger = 1..n;

int F[forretninger] = ...;
int c[forretninger][gaver] = ...;


dvar boolean z[forretninger];
dvar boolean w[forretninger][gaver];


minimize sum ( f in forretninger , g in gaver ) c[f][g]*w[f][g] + sum ( f in forretninger ) F[f]* z[f];
subject to
{
  forall ( g in gaver )
  {
    sum ( f in forretninger ) w[f][g] == 1;
    forall ( f in forretninger )
    {
        w[f][g] - z[f] <= 0;
    }
  }
}
```

Med følgende data-fil

```
n = 6;
m = 15;
F = [150 65 80 100 25 100];
c = [   [100000 100000  550 100000  400 400 400 450 100000  400 100000  500 450 350 400],
        [400    500 100000  650 100000  400 100000  350 600 300 400 550 400 450 500],
        [100000 600 550 100000  250 450 400 450 600 200 350 100000  100000  100000  600],
        [550    100000  550 100000  350 100000  100000  550 600 100000  450 600 400 500 650],
        [650    650 600 100000  100000  100000  450 500 100000  100000  100000  500 300 600 700],
        [100000 700 100000  550 300 600 100000  100000  100000  250 400 550 300 500 650]];

```

En optimal løsning koster Beate 6.495 kroner og hun skal benytte forretningerne 1, 2, 3 og 5. Gaverne skal købes ved følgende forretninger

Gave	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15
Forretning	2	2	1	6	3	1	1	2	2	3	3	1	6	1	1

Samlet set skal Beate betale 6.100 kroner for gaverne og 395 for forsendelse. Sammenlignes løsningen med den fra Opgave 1, ses at de direkte omkostningen til køb af gaver forbliver uændret, men vi sparer 420-395=25 kroner i fragtomkostninger. Dette opnås ved at købe gave nummer 13 fra forretning 6 i stedet for fra forretning 5.

Det går nu op for Beate, at der ved alle de online forretninger hun har udvalgt sig er en regel der siger, at der er fri fragt, hvis man køber for 500 kroner eller mere.

Opgave 7

Formuler, med udgangspunkt i modellen fra Opgave 4, en ny model, som tager højde for, at Beate opnår "fri fragt" hvis hun køber for 500 kroner eller mere ved en forretning.

Svar: Vi indfører en ny binær variabler y_f for hver forretning $f = 1, \dots, 6$ givet ved

$$y_f = \begin{cases} 1, \mathit{hvis\ Beate\ køber\ for\ } 500 \mathit{\ kroner\ eller\ mere\ ved\ forretning\ } f = 1, \dots, 6 \\ 0, \mathit{ellers} \end{cases}$$

Det første vi således skal gøre er, at udregne præcis hvor mange penge Beate bruger ved hver forretning. Ved forretning f bruger Beate $\sum_{g=1}^{15} c_{fg} w_{fg}$ kroner. Det vil sige, at vi ønsker at modellere den logiske implikation

$$y_f = 1 \Rightarrow \sum_{g=1}^{15} c_{fg} w_{fg} \geq 500$$

Dette kan gøres ved hjælp af følgende lineære uligheder:

$$\sum_{g=1}^{15} c_{fg} w_{fg} \geq 500 y_f, \quad \forall f = 1, \dots, 6$$

Da der ikke eksplicit står i opgaven at modellen skal være en lineær heltalsmodel, er det også i orden blot at give implikationen

$$y_f = 1 \Rightarrow \sum_{g=1}^{15} c_{fg} w_{fg} \geq 500$$

som begrænsninger, da dette kan implementeres direkte i OPL vha. indikator-begrænsninger.

Yderligere skal objektfunktionen udvides så den bliver

$$\min \sum_{f=1}^{6} \sum_{g=1}^{15} c_{fg} w_{fg} + \sum_{g=1}^{6} F_f (z_f - y_f)$$

Hvorved den fragtomkostningerne bortfalder hvis der handles for mere end 500 kroner. Den samlede model bliver

$$\min \sum_{f=1}^{6}\sum_{g=1}^{15} c_{fg}w_{fg} + \sum_{g=1}^{6} F_f(z_f - y_f)$$

$$\text{s.t.:} \sum_{f=1}^{6} w_{fg} = 1, \quad \forall g = 1, \dots, 15$$

$$w_{fg} \leq z_f, \quad \forall f = 1, \dots, 6 \; og \; g = 1, \dots, 15$$

$$\sum_{g=1}^{15} c_{fg}w_{fg} \geq 500y_f, \quad \forall f = 1, \dots, 6$$

$$\left(y_f = 1 \Rightarrow \sum_{g=1}^{15} c_{fg}w_{fg} \geq 500, \quad \forall f = 1, \dots, 6 \right)$$

$$w_{fg} \in \{0,1\}, \quad \forall f = 1, \dots, 6 \; og \; g = 1, \dots 15$$

$$z_f \in \{0,1\}, \quad \forall f = 1, \dots, 6$$

$$y_f \in \{0,1\}, \quad \forall f = 1, \dots, 6$$

Opgave 8:

Implementer modellen fra Opgave 7 i OPL og løs den ved hjælp af CPLEX. Dokumentér din implementering for eksempel ved hjælp af screenshots, kommentér på løsningen og sammenlign den med løsningen fundet i Opgave 6.

Svar: En implementering af modellen fra Opgave 8 kan se ud som følger

```
int n = ...;
int m = ...;
range gaver = 1..m;
range forretninger = 1..n;

int F[forretninger] = ...;
int c[forretninger][gaver] = ...;


dvar boolean z[forretninger];
dvar boolean w[forretninger][gaver];
dvar boolean y[forretninger];


minimize sum ( f in forretninger , g in gaver ) c[f][g]*w[f][g] + sum ( f in forretninger ) F[f]* ( z[f] - y[f]);
subject to
{
  forall ( g in gaver )
  {
    sum ( f in forretninger ) w[f][g] == 1;
    forall ( f in forretninger )
    {
        w[f][g] - z[f] <= 0;
    }
  }
  forall ( f in forretninger )
  {
    sum ( g in gaver ) c[f][g] * w[f][g] >= 500*y[f];
  }
}
```

Her er datafilen ikke ændret fra Opgave 6. En optimal løsning koster Beate 6.100kroner og hun skal benytte forretningerne 1, 2, 3, 5 og 6. Hun opnår nu helt at undgå at betale fragt da de samlede fragtomkostninger beløber sig til 0kroner. Hun køber således for følgende beløb hos de forskellige forretninger

Forretning	Beløb
1	2100
2	1850
3	800
4	0
5	800
6	550

Dermed opnås altså fri fragt hos alle forretninger og hun skal kun betale de direkte omkostninger til gaverne som beløber sig til 6.100 kroner. Der er mere end én optimal løsning til dette program.